
Lists
Commands:

→



Used for storing to a list
dim(



Used for storing or getting dimension of a list
Fill(



Stores a value to all elements of a list
SortA(



Sorts elements of a list in ascending order

SortD(



Sorts elements of a list in descending order

augment(


Concatenates two lists
Using the Commands:

*Note:  These tutorials will use [L] for the small L list symbol.  This symbol can be found by pressing [2nd], [STAT], [>], and then scrolling all the way down, or can be accessed via the Catalog.
Lists are generally used for storing large amounts of data or for creating save systems in games.  There are several commands for manipulating data stored within lists, sometimes making it advantageous to use lists rather than variables to store data.  You can use one of the calculator’s six default lists, but generally it is better to create your own.  To create a list, you can either directly store some data to it, or you can define it by defining its dimension:
value→dim([L]listname

where value is the number of elements you want in the list and listname is the name of your list, which can consist of up to five characters.  dim( can also be used to return the dimension of a list, which can be useful when using a For( loop to access data:
dim([L]listname
You can store a specific value to every element of an *existing* list by using Fill(:

Fill(value,listname

You can sort the contents of a list in either ascending or descending order using SortA( and SortD(:

SortA(listname

SortD(listname
You can concatenate, or combine, two lists by using augment(:

augment(listname1,listname2

It is important to note that augment( only returns a concatenation of the two lists.  It is your job, then, to use this returned list by doing something like using Ans or storing it to a third list:
augment(listname1,listname2→listname3

To return a specific element of a list, do this:

listname(element)

where listname is the name of the list and element is the location of an element in the list.
Sample code:




Output:
[image: image1.png]@2t




:ClrHome

:4→dim([L]ABC

:Disp [L]ABC

:Pause

:Fill(7,[L]ABC

:Disp [L]ABC

:Pause

:ClrHome

Explanation:

First, the program defines the list ABC by storing its dimension.  Since 4 is stored to the dimension of the list, it will have four elements.  Next, the program displays the contents of the list.  The first time you run this program, all the list elements should be 0.  After pausing, the program uses Fill( to store 7 to every element in the list, then displays the contents of the list again.  Now, run the program a second time.  Notice that instead of all the elements being 0, they are all 7.  The beauty of dim( is that it allows a list to be defined, but it doesn’t affect the data stored in the list (unless you cut off elements by storing a smaller dimension to the list).

Sample code:




Output:

:ClrHome

:Input “DIMENSION: ”,A

:A→dim([L]TEST

:For(Z,1,dim([L]TEST

:Input “VALUE: ”,V
:V→[L]TEST(Z

:End

:Pause [L]TEST

:SortA([L]TEST

:Pause [L]TEST

:SortD([L]TEST

:Pause([L]TEST

Explanation:

First the program asks you for the dimension of the list.  Next a For( loop asks you for a number of elements depending on the dimension and stores them into the list.  After all the elements are entered, Pause is used to display the list.  The list is then sorted in ascending order, displayed, sorted in descending order, and displayed again.  This program uses Pause in an interesting way that we haven’t gone over before, which applies to displaying text as well as to displaying lists.  Really, you can almost always use just Pause instead of Disp and Pause.  For example, instead of:

:Disp “HELLO

:Pause

you could just do this:

:Pause “HELLO

It does exactly the same thing and saves memory.  In our program here it is actually much better to use Pause to display our list because it allows us to scroll left and right through the list if it goes off the screen.  If we were to use Disp and then Pause, the list might just go off the screen and you wouldn’t be able to scroll to see all its elements.
Sample code:




Output:

:ClrHome

:{1,2,3→[L]DAT1

:{4,5,6→[L]DAT2

:Pause [L]DAT1

:Pause [L]DAT2

:augment([L]DAT1,[L]DAT2→[L]DAT3

:Pause [L]DAT3

Explanation:

The program first stores data to two different lists and displays their contents.  Then the lists are concatenated and the result is displayed.  In the next sample code we will learn how to do something more useful: using lists to create a save system for a game.

Sample code:
:ClrHome

:AxesOff:FnOff

:5→dim([L]GAME

:Menu(“DO WHAT?”,“CONTINUE”,1,“NEW GAME”,2

This block of code is executed if “CONTINUE” is selected.  If the list indicates that saved data is present, the saved data is stored to the variables.  Otherwise, the variables are reset.
:Lbl 1

:If [L]GAME(1)=47:Then
:[L]GAME(2)→A:[L]GAME(3)→B:[L]GAME(4)→C:[L]GAME(5)→D
:0→[L]GAME(1)
:Else
:ClrDraw

:Text(0,0,“NO  SAVED  FILE  IN  MEMORY

:Pause:ClrDraw
:DelVar A:DelVar B:DelVar C:DelVar D
:End

:Goto 3

This block of code is executed if “NEW GAME” is selected.  The variables are reset.
:Lbl 2

:DelVar A:DelVar B:DelVar C:DelVar D

Main menu
:Lbl 3
:ClrHome
:Menu(“MAIN MENU”,“CHANGE DATA”,4,“VIEW DATA”,5,“QUIT”,6

Allows user to change data stored in variables
:Lbl 4
:ClrHome

:Input “A: ”,A

:Input “B: ”,B

:Input “C: ”,C

:Input “D: ”,D

:ClrHome

:Goto 3

Displays data stored in the variables
:Lbl 5

:ClrDraw

:Text(0,0,“A=”,A

:Text(7,0,“B=”,B

:Text(14,0,“C=”,C

:Text(21,0,“D=”,D

:Pause:ClrDraw

:Goto 3

Saves data to a list, then program ends
:Lbl 6

:47→[L]GAME(1)

:A→[L]GAME(2):B→[L]GAME(3):C→[L]GAME(4):D→[L]GAME(5)

Explanation:

First of all, I hope you noticed that comments are in blue and should not be entered into the program editor.  I also divided the code into blocks so it’s easier to understand, but you don’t have to do that when you enter the program unless you want to.

This program is a good example of how you can use lists as a save system for a game.  The first element in the list is just an arbitrary number that is stored there when the data is saved upon exiting the program.  It basically allows the program to check to see if data has been previously saved to the list.  Pretty ingenious, huh?  If “CONTINUE” is selected from the first menu and the first list element indicates that the list contains saved data, the program stores the saved data from the list to variables.  Then the first list element is set to zero to prevent cheating (like exiting the program by pressing [ON] after losing to avoid resaving new data).  If “CONTINUE” is selected, but the first list element indicates that no data has been previously saved, the variables are reset.  The variables are also reset if “NEW GAME” is selected.

Your main program should start where label 3 is.  Here, I just put in an option to manipulate the variables and to view the data stored in the variables.  In a real game, the “CHANGE DATA” option could be replaced with a shop and/or the main gameplay function where data could potentially be altered.  If you’re making a text-based game, it is generally a good idea to have an inventory screen (similar to the “VIEW DATA” option) so that the user can keep track of their stats.

When “QUIT” is selected, the program automatically saves the variables to the list.  A value (in this case 47) is stored to the first element of the list to indicate that data has been saved.
Since it is still possible to alter data contained within the list from outside of the program, you may want to take some steps to prevent cheating.  Instead of storing the value that indicates saved data exists to the first element, try storing it to an element somewhere in the middle of the list.  You can also do things like store the fourth element in the list to a variable first, then the seventh element, etc. so that if someone looks at your sourcecode, it is hard to figure out which list element corresponds to which statistic in your game.

Conclusion:

Lists can be very useful for both storing data and manipulating it.  You may never (or very rarely) use some of the manipulation commands like SortA( or augment(, but if you find yourself in a situation where you need to manipulate a bunch of data, they could prove invaluable.  Remember to define a list before trying to access data from it!  All this takes is something like:

:value→dim([L]listname

Next, go to “Matrices”.

