
Game: Falling Pixel

Introduction:
Finally, we are ready to make a game.  In this game, you are a pixel that starts at the top of the screen and you must work your way to the bottom while avoiding other pixels in order to move to the next level.  Create a new program a name it Fall83P if you have a TI-83 Plus or Fall84P if you have a TI-84 Plus.  Then enter the code and execute the program.  Comments are in blue and should not be entered into your program.  And if you have an 84+, there are a few very minor changes you should make as directed in the comments.
::"FALLING  PIXEL  v1.0/83+
For use with shells.  84+: change “83+” to “84+”
:ClrDraw:AxesOff:FnOff 

:0→Xmin:94→Xmax


Setting up our graphscreen…
:-62→Ymin:0→YMax

:Text(0,23,"FALLING  PIXEL
Title screen
:Text(20,23,"PROGRAMMED  BY

:Text(27,29,"MDR  FALCON

:Text(55,41,"v1.0

:Pause :ClrDraw

:100→C:1→L:0→S


100→# of pixels on screen; 1→level; 0→score
:Lbl 1

:S+L^2→S



Scoring bonus for completing levels
:Text(24,32,"LEVEL  ",L

Tells us what level we’re on
:Pause :ClrDraw

:Vertical 0:Vertical 94


Draws left and right boundaries
:If C=20:RecallPic 0


If after the first level, recall pic from previous level
:For(Z,1,C



Add more pixels to the screen
:randInt(5,62→A

:randInt(1,93→B

:Pxl-On(A,B

:End

:StorePic 0



Store the gameplay screen for future use
:Text(20,27,"PRESS  ENTER

:Text(27,33,"TO  BEGIN

:Pause 

:ClrDraw

:RecallPic 0

:0→A:47→B



Sets initial coordinates of the falling pixel
:Repeat pxl-Test(A,B


Gameplay starts here.  If pixel is hit, you lose.
:Pxl-On(A,B



Turns on falling pixel
:If A=62:Goto 2


If falling pixel is at the bottom, you win the level
:For(Z,1,15



Waste time.  84+: change to For(Z,1,40
:End




End time-wasting loop
:S+1→S



Add 1 to score
:Pxl-Off(A,B



Turns off falling pixel so it can move
:getKey→K

:B+(K=26)-(K=24→B

Moves pixel horizontally if arrow key was pressed
:A+1→A



Falling pixel falls
:End




End of main gameplay loop
:ClrDraw

:Text(24,32,"YOU  LOSE!

You lost
:Text(56,0,"SCORE: ",S

Display score
:Pause :ClrDraw

:Return




Stop the program
:Lbl 2

:Text(24,31,"COMPLETE!

Level is complete
:Pause :ClrDraw

:20→C




20 new pixels will be drawn each level
:L+1→L



Increase level by one
:Goto 1



Jump to label 1
Explanation:

Alright… so the very first line of code is in case you want the program to appear in a shell such as MirageOS.  The colon is what actually makes it show up in the shell; the text allows MirageOS to display the title of your program at the bottom of the screen.  Next, we have our lovely title screen, and yes, the credit should be given to me, MDR Falcon, because I wrote the program and have already uploaded it to various calc sites.  

After the title screen, we store 100 to variable C, 1 to L, and 0 to S.  C is used for the number of new pixels that will appear in the For( loop that comes a little later.  The initial value of C is 100 because we want 100 pixels to appear for the first level.  Later, 20 will be stored to C because we will want 20 new pixels to appear on each level after the first one.  Of course, L is given an initial value of 1 because you start on level 1.  Variable S is used to keep track of scoring, so its initial value is set to 0.

Next, we have label 1.  After a level is completed, the program will jump back to here so that a new level can be started.   After label 1, we have an expression that adds a bonus to the score for beating levels.  There is a screen to tell you which level you’re on, and then the gameplay screen is set up.  A vertical line is drawn on each side on the screen and then a For( loop is used to draw a number of random pixels based on the value of C.  Right before the For( loop, there is a line that says to recall Pic 0 if C=20.  When C=20, it means that we are on at least the second level because 20 is stored to C after the end of the first level.  It makes for much less waiting if we just use the pixels from the previous levels and add 20 more than if we were to draw over a hundred pixels at the beginning of each level.  After the new pixels are drawn, the program stores the screen to Pic 0 so it can be used later.  Next, some text pops up to tell the user to press [ENTER] so that they are not caught off guard when the game starts.
Finally, we are to the main gameplay loop.  Before the loop begins, the initial coordinates of our falling pixel are stored to A and B.  For our loop, we are using Repeat with a pixel-test.  Basically what this does is repeat the loop until the falling pixel hits another pixel.  The first thing that happens in the loop is that the falling pixel is turned on.  If the pixel is at the bottom of the screen (A=62), then the program jumps down to label 2.  There is a For( loop to waste time (which is a rare thing to do in a BASIC game) and then the score is incremented.  Next the falling pixel is turned off.  If the left or right arrow key is pressed, the horizontal coordinate of the falling pixel is incremented or decremented.  Variable A is incremented by 1, which makes the pixel fall, and the loop ends and starts over.  It is important that the falling pixel is turned off before its coordinates are changed or else a trail of pixels will be left behind it.

The five lines of code after the end of the loop are executed only if a pixel is hit.  Why?  Because 1.)the loop ends on the condition that a pixel was hit, and 2.)if the falling pixel was at the bottom of the screen indicating that it was the end of the level, the program would have just jumped to label 2.  So if a pixel was hit, there is a screen telling you that you lost and what your score is.  Then a Return command is used to end the program.  Return is almost exactly like Stop, though I hear that Return is better to use if you are executing the program from a shell or another program.
Label 2… The program jumps here from the loop when the level is completed.  There is a screen that tells you that the level is complete, then the program jumps back to label 1.  Be aware that it is generally not good to jump out of loops using Goto if it can be avoided because it causes memory leaks.  Here, it is kind of hard to avoid and won’t affect gameplay unless you are really low on RAM.
Conclusion:

Ah… a full page of explanation.  I have trouble writing that much on English papers.  You should be able to understand most of what’s going on just by looking at the code and reading the comments.  If you’re having trouble, execute the program and follow the source code line by line.  If you get it, you should especially take time to analyze how we used the Repeat loop, getKey, and pictures.  Next, go to “Lists”.

